# Maintainer: Julien Nicoulaud <julien.nicoulaud@gmail.com>
# Contributor: James Rayner <james@archlinux.org>
# Contributor: Todd Maynard <arch@toddmaynard.com>
# Source: https://github.com/nicoulaj/archlinux-packages
pkgname=python2-cheetah
pkgver=2.4.4
pkgrel=5
pkgdesc="A Python-powered template engine and code generator"
arch=(i686 x86_64)
url="http://www.cheetahtemplate.org"
license=(custom)
depends=(python2 python2-markdown)
makedepends=(python2-distribute)
optdepends=('python2-pygments: for the CodeHighlighter filter')
provides=(cheetah python-cheetah)
conflicts=(cheetah python-cheetah)
changelog=Changelog
source=(http://pypi.python.org/packages/source/C/Cheetah/Cheetah-${pkgver}.tar.gz)
md5sums=('853917116e731afbc8c8a43c37e6ddba')

build() {
  cd "${srcdir}/Cheetah-${pkgver}"
  python2 setup.py build
}

package() {
  msg "Install..."
  cd "${srcdir}/Cheetah-${pkgver}"
  python2 setup.py install --root="${pkgdir}" || return 1

  msg2 "Install copyright resources in /usr/share/licenses/${pkgname}..."
  install -Dm644 "${srcdir}/Cheetah-${pkgver}/LICENSE" "${pkgdir}/usr/share/licenses/${pkgname}/LICENSE"
}
